

Das Modul Student:-Calculus1

Zum Selberrnachen:

```
[ > #?Student:-Calculus1
```

1. Schrittweises Lösen von Problemen

```
[ > restart:  
with(Student:-Calculus1):
```

a) Probleme und Regeln (Limit, Diff, Int und Rule)

Wir setzen das `infolevel` auf 1, damit wir einige wenige zusätzliche Ausgaben über die Verwaltung von Problemen sowie Erklärungen zu komplizierten Tipps (siehe b) erhalten:

```
> infolevel[Calculus1] := 1:
```

Probleme werden mit den *inerten* Befehlen `Limit`, `Diff` und `Int` definiert:

```
> Limit(x^2, x=3);
```

$$\lim_{x \rightarrow 3} x^2$$

Rechenregeln werden mit dem Befehl `Rule` angewandt, der als *Index* in eckigen Klammern die Regel erhält, als *Argument* in runden Klammern das Problem:

```
> Rule[power](%);
```

```
Creating problem #1
```

$$\lim_{x \rightarrow 3} x^2 = \left(\lim_{x \rightarrow 3} x \right)^2$$

Die Ausgabe eines Aufrufs von `Rule` ist normalerweise eine Gleichung, wobei die linke Seite das ursprüngliche Problem ist, und die rechte Seite der aktuelle Ausdruck. Auf diese Ausgabe kann `Rule` erneut angewendet werden.

```
> Rule[identity](%);
```

$$\lim_{x \rightarrow 3} x^2 = 9$$

Der Befehl `ShowSteps` liefert die Schritte der Rechnung als Gleichungskette:

```
> ShowSteps();
```

$$\begin{aligned} \lim_{x \rightarrow 3} x^2 &= \left(\lim_{x \rightarrow 3} x \right)^2 \\ &= 9 \end{aligned}$$

Ein Differentiationsproblem:

```
> Rule[product](Diff(x^2*sin(x), x));
```

```
Creating problem #2
```

$$\frac{d}{dx} (x^2 \sin(x)) = \left(\frac{d}{dx} (x^2) \right) \sin(x) + x^2 \left(\frac{d}{dx} \sin(x) \right)$$

Die grundlegenden Regeln haben Kurzformen: `difference = ` - ``, `product = ` * ``, `power = ` ^ ``, `quotient = ` / ``, `constantmultiple = ` c * ``. Zur Vollständigkeit gibt es auch `sum = ` + ``, obwohl beide Befehle drei Zeichen lang sind.

```
> Rule[^^](%);
```

$$\frac{d}{dx} (x^2 \sin(x)) = 2 x \sin(x) + x^2 \left(\frac{d}{dx} \sin(x) \right)$$

Jede univariate Funktion kann als Regel verwendet werden:

```
> Rule[sin](%);
```

$$\frac{d}{dx} (x^2 \sin(x)) = 2 x \sin(x) + x^2 \cos(x)$$

Das Paket speichert alle Probleme. Man kann mit dem Befehl `GetProblem` auf ein vorher behandeltes Problem zurückgreifen:

```
> GetProblem(1);
```

$$\lim_{x \rightarrow 3} x^2 = 9$$

Man kann den aktuellen Stand eines Problem mit dem Befehl `Show` anzeigen lassen, oder auch den aller Probleme:

```
> Show(2);
```

$$\frac{d}{dx} (x^2 \sin(x)) = 2 x \sin(x) + x^2 \cos(x)$$

```
> Show(all);
```

```
Problem 1:
```

$$\lim_{x \rightarrow 3} x^2 = 9$$

Problem 2:

$$\frac{d}{dx}(x^2 \sin(x)) = 2x \sin(x) + x^2 \cos(x)$$

b) Tipps (Hint)

> `Diff(sin(x^2),x);`

$$\frac{d}{dx} \sin(x^2)$$

Der Befehl `Hint` gibt einen Tipp zu einem Problem in Form einer anwendbaren Regel:

> `Hint(%);`

Creating problem #3

[chain]

Die Ausgabe von `Hint` kann grundsätzlich als Index für einen Aufruf von `Rule` dienen:

> `Rule[%](%%);`

$$\frac{d}{dx} \sin(x^2) = \left(\frac{d}{d_X} \sin(_X) \right) \Big|_{_X=x^2} \left(\frac{d}{dx} (x^2) \right)$$

> `Int(x*cos(x),x);`

$$\int x \cos(x) dx$$

> `Hint(%);`

Creating problem #4

[parts, x, sin(x)]

> `Rule[%](%%);`

$$\int x \cos(x) dx = x \sin(x) - \int \sin(x) dx$$

> `Diff(x^x,x);`

$$\frac{d}{dx} x^x$$

> `Hint(%);`

Creating problem #5

Rewrite the expression as an exponential

[rewrite, x^x = e^{(x ln(x))}]

> `Rule[%](%%);`

$$\frac{d}{dx} x^x = \frac{d}{dx} (e^{(x \ln(x))})$$

> `Rule[chain](%);`

$$\frac{d}{dx} x^x = \left(\frac{d}{d_X} (e^X) \right) \Big|_{_X=x \ln(x)} \left(\frac{d}{dx} (x \ln(x)) \right)$$

> `Rule[exp](%);`

$$\frac{d}{dx} x^x = e^{(x \ln(x))} \left(\frac{d}{dx} (x \ln(x)) \right)$$

> `Rule[product](%);`

$$\frac{d}{dx} x^x = e^{(x \ln(x))} \left(\left(\frac{d}{dx} x \right) \ln(x) + x \left(\frac{d}{dx} \ln(x) \right) \right)$$

> `Rule[power](%);`

$$\frac{d}{dx} x^x = e^{(x \ln(x))} \left(\ln(x) + x \left(\frac{d}{dx} \ln(x) \right) \right)$$

> `Rule[ln](%);`

$$\frac{d}{dx} x^x = e^{(x \ln(x))} (\ln(x) + 1)$$

> `Hint(%);`

This problem is complete

```
[ ]
[ ] > simplify(%);
```

$$\frac{d}{dx} x^x = x^x (\ln(x) + 1)$$

c) Eine Mogelprozedur mittels `Hint` und `ShowSteps`

Mit Hilfe der vorgestellten Befehle lässt sich jetzt leicht eine Prozedur schreiben, die Probleme löst und dabei alle Zwischenschritte ausgibt, was Maple normalerweise nicht macht:

```
> mogel := proc( P )
  local prob, hint;
  prob := P;
  hint := Hint(prob);
  while nops(hint) > 0 do
    prob := Rule[hint](prob);
    hint := Hint(prob);
  end do;
  ShowSteps();
end proc;
```

```
> mogel( Limit(x^x, x=0, right) );
Creating problem #6
```

Rewrite the expression as an exponential to prepare for using l'Hopital's rule
This problem is complete

$$\begin{aligned} \lim_{x \rightarrow 0^+} x^x &= \lim_{x \rightarrow 0^+} e^{(x \ln(x))} \\ &= e^{\left(\lim_{x \rightarrow 0^+} x \ln(x) \right)} \\ &= e^{\left(\lim_{x \rightarrow 0^+} -x \right)} \\ &= e^{\left(-\left(\lim_{x \rightarrow 0^+} x \right) \right)} \\ &= e^{-0} \\ &= 1 \end{aligned}$$

d) Einfache Regeln automatisch anwenden (`Understand`)

Das erste Argument von `Understand` ist der Typ des Problems; die restlichen Argumente sind die Regeln, die automatisch angewendet werden sollen.

```
> Understand(Diff, constant, `c*`, `+`);
```

`Diff = [constant, constantmultiple, sum]`

Nachdem die Produktregel angewendet wurde, wendet das System automatisch alle *verstandenen* Regeln an.

```
> Rule[ `*` ](Diff(x^2*(cos(x)-2*ln(x)), x));
Creating problem #7
```

$$\frac{d}{dx} (x^2 (\cos(x) - 2 \ln(x))) = \left(\frac{d}{dx} (x^2) \right) (\cos(x) - 2 \ln(x)) + x^2 \left(\left(\frac{d}{dx} \cos(x) \right) - 2 \left(\frac{d}{dx} \ln(x) \right) \right)$$

Um alle verstandenen Regeln anzuwenden, benutze einen leeren Index für den Befehl `Rule`.

```
> Rule[](Diff(3*x+exp(x), x));
Creating problem #8
```

$$\frac{d}{dx} (3x + e^x) = 3 \left(\frac{d}{dx} x \right) + \left(\frac{d}{dx} (e^x) \right)$$

e) Ein komplizierteres Beispiel (`ShowIncomplete`)

Wir betrachten ein Integrationsproblem:

```
> Int(w^(1/3)/(w^(1/2)+1), w=a..b);
```

$$\int_a^b \frac{w^{(1/3)}}{\sqrt{w+1}} dw$$

Maple gibt folgenden Tipp: Substituiere $w = u^6$. Das ist naheliegend, weil sich dann bequem sowohl die dritte als auch die Quadratwurzel von w ziehen lassen.

> **Hint(%);**
 Creating problem #9

[change, w = u⁶, u]

Bei dieser Substitution müssen eigentlich die Grenzen unseres bestimmten Ideals angepasst werden. In diesem Fall sind die neuen Grenzen a^(1/6) und b^(1/6). Da hier Wurzeln involviert sind, gibt Maple die Grenzen weiter in der ursprünglichen Variablen an.

> **Rule[%](%%);**
 Applying substitution w = u⁶, u = w^(1/6) with dw = 6*u⁵*du, du = 1/6/w^(5/6)*dw

$$\int_a^b \frac{w^{(1/3)}}{\sqrt{w+1}} dw = \int_{w=a}^{w=b} 6u^4 - 6u + \frac{6u}{u^3+1} du$$

> **Understand(int, constant, `c*`, `+`);**
 Int = [constant, constantmultiple, sum]

> **Rule[](%%);**

$$\int_a^b \frac{w^{(1/3)}}{\sqrt{w+1}} dw = 6 \int_{w=a}^{w=b} u^4 du - 6 \int_{w=a}^{w=b} u du + 6 \int_{w=a}^{w=b} \frac{u}{u^3+1} du$$

> **Rule[Hint(%)](%%);**

$$\int_a^b \frac{w^{(1/3)}}{\sqrt{w+1}} dw = 6 \left(\frac{u^5}{5} \right)_{w=a..b} - 6 \int_{w=a}^{w=b} u du + 6 \int_{w=a}^{w=b} \frac{u}{u^3+1} du$$

> **Rule[Hint(%)](%%);**

$$\int_a^b \frac{w^{(1/3)}}{\sqrt{w+1}} dw = 6 \left(\frac{u^5}{5} \right)_{w=a..b} - 6 \left(\frac{u^2}{2} \right)_{w=a..b} + 6 \int_{w=a}^{w=b} \frac{u}{u^3+1} du$$

> **Hint(%);**
 [partialfractions]

> **Rule[%](%%);**

$$\int_a^b \frac{w^{(1/3)}}{\sqrt{w+1}} dw = 6 \left(\frac{u^5}{5} \right)_{w=a..b} - 6 \left(\frac{u^2}{2} \right)_{w=a..b} + 2 \int_{w=a}^{w=b} \frac{u+1}{u^2-u+1} du - 2 \int_{w=a}^{w=b} \frac{1}{u+1} du$$

Zu diesem Zeitpunkt gibt es noch zwei (Integrations-)Unterprobleme. Um mit einem bestimmten dieser Unterprobleme arbeiten zu können, z.B. mit dem komplizierteren, können wir uns ein sog. *Label* mit Hilfe des Befehls **ShowIncomplete** geben lassen und dieses bei Aufrufen von **Rule** oder **Hint** verwenden.

> **ShowIncomplete();**

$$\%Int14 = \int_{w=a}^{w=b} \frac{u+1}{u^2-u+1} du$$

$$\%Int15 = \int_{w=a}^{w=b} \frac{1}{u+1} du$$

> **lbl := %Int15;**
 lbl := %Int15

Wenn der Tipp kompliziert ist, wird bei **infolevel[Calculus1]** oder **infolevel[Student] >= 1** eine Nachricht mit zusätzlichen Informationen ausgegeben. Das haben wir z.T. schon oben gesehen.

> **Hint(lbl);**
 [change, ul = u + 1, ul]

Achtung: Die Labels der Unterprobleme werden vom Modul **Calculus1** verwaltet und können sich zwischen zwei Anwendungen von **Rule** ändern. Daher sollte man, bevor man sich mit einem Label auf ein Unterproblem bezieht, durch **ShowIncomplete** davon überzeugen, dass das Label korrekt ist.

> **Rule[%](lbl);**

Applying substitution $u = u_1 - 1$ with $du = du_1$

$$\int_{w=a}^{w=b} \frac{1}{u+1} du = \int_{w=a}^{w=b} \frac{1}{u_1} du_1$$

> **Hint(%);**

[power]

> **Rule[%](%%);**

$$\int_{w=a}^{w=b} \frac{1}{u+1} du = \ln(u_1) \Big|_{w=a..b}$$

> **Rule[Hint(%)](%);**

Reverting substitution using $u_1 = u+1$

$$\int_{w=a}^{w=b} \frac{u+1}{3(u^2-u+1)} - \frac{1}{3(u+1)} du = \frac{1}{3} \int_{w=a}^{w=b} \frac{u+1}{u^2-u+1} du - \frac{1}{3} \ln(u+1) \Big|_{w=a..b}$$

Da die Variable u_1 in keinem ungelösten Unterproblem vorkommt, mache die zugehörige Substitution rückgängig (*revert*).

> **Hint(%);**

Rewrite the numerator in a form which contains the derivative of the denominator

$$\left[\text{rewrite}, \frac{u+1}{u^2-u+1} = \frac{2u-1}{2(u^2-u+1)} + \frac{3}{2(u^2-u+1)} \right]$$

> **Rule[%](%%);**

$$\int_{w=a}^{w=b} \frac{u+1}{3(u^2-u+1)} - \frac{1}{3(u+1)} du = \frac{1}{6} \int_{w=a}^{w=b} \frac{2u-1}{u^2-u+1} du + \frac{1}{2} \int_{w=a}^{w=b} \frac{1}{u^2-u+1} du - \frac{1}{3} \ln(u+1) \Big|_{w=a..b}$$

> **Hint(%);**

Note that the derivative of u^2-u+1 is $2u-1$, so we can make a change of variable.

[change, $u_1 = u^2 - u + 1, u_1$]

> **Rule[%](%%);**

Applying substitution $u = 1/2 + 1/2 * (-3 + 4 * u_1)^{1/2}$, $u_1 = u^2 - u + 1$ with $du = 1 / (-3 + 4 * u_1)^{1/2} * du_1$, $du_1 = (2 * u - 1) * du$

$$\int_{w=a}^{w=b} \frac{u+1}{3(u^2-u+1)} - \frac{1}{3(u+1)} du = \frac{1}{6} \int_{w=a}^{w=b} \frac{1}{u_1} du_1 + \frac{1}{2} \int_{w=a}^{w=b} \frac{1}{u^2-u+1} du - \frac{1}{3} \ln(u+1) \Big|_{w=a..b}$$

> **Rule[Hint(%)](%);**

$$\int_{w=a}^{w=b} \frac{u+1}{3(u^2-u+1)} - \frac{1}{3(u+1)} du = \frac{1}{6} \ln(u_1) \Big|_{w=a..b} + \frac{1}{2} \int_{w=a}^{w=b} \frac{1}{u^2-u+1} du - \frac{1}{3} \ln(u+1) \Big|_{w=a..b}$$

> **Rule[revert](%);**

Reverting substitution using $u_1 = u^2 - u + 1$

$$\int_{w=a}^{w=b} \frac{u+1}{3(u^2-u+1)} - \frac{1}{3(u+1)} du = \frac{1}{6} \ln(u^2-u+1) \Big|_{w=a..b} + \frac{1}{2} \int_{w=a}^{w=b} \frac{1}{u^2-u+1} du - \frac{1}{3} \ln(u+1) \Big|_{w=a..b}$$

Nach dem Rücksubstituieren von u_1 ist das Unterproblem gelöst. Maple setzt das Ergebnis in das nächstgrößere Problem ein.

> **Rule[revert](%);**

Rule [revert] does not apply

$$\int_{w=a}^{w=b} \frac{u+1}{3(u^2-u+1)} - \frac{1}{3(u+1)} du = \frac{1}{6} \ln(u^2-u+1) \Big|_{w=a..b} + \frac{1}{2} \int_{w=a}^{w=b} \frac{1}{u^2-u+1} du - \frac{1}{3} \ln(u+1) \Big|_{w=a..b}$$

> **Rule[change, u_1=u+1](%);**

Applying substitution $u = u_1 - 1$ with $du = du_1$

$$\int_{w=a}^{w=b} \frac{u+1}{3(u^2-u+1)} - \frac{1}{3(u+1)} du = \frac{1}{6} \ln(u^2-u+1) \Big|_{w=a..b} + \frac{1}{2} \int_{w=a}^{w=b} \frac{1}{-3u_1 + u_1^2 + 3} du_1 - \frac{1}{3} \ln(u+1) \Big|_{w=a..b}$$

```
> Rule[^^](%);
```

```
Rule [power] does not apply
```

$$\int_{w=a}^{w=b} \frac{u+1}{3(u^2-u+1)} - \frac{1}{3(u+1)} du = \frac{1}{6} \ln(u^2-u+1) \Big|_{w=a..b} + \frac{1}{2} \int_{w=a}^{w=b} \frac{1}{-3ul+ul^2+3} du - \frac{1}{3} \ln(u+1) \Big|_{w=a..b}$$

```
> Rule[revert](%);
```

```
Rule [revert] does not apply
```

$$\int_{w=a}^{w=b} \frac{u+1}{3(u^2-u+1)} - \frac{1}{3(u+1)} du = \frac{1}{6} \ln(u^2-u+1) \Big|_{w=a..b} + \frac{1}{2} \int_{w=a}^{w=b} \frac{1}{-3ul+ul^2+3} du - \frac{1}{3} \ln(u+1) \Big|_{w=a..b}$$

Eine letzte Rücksubstitution und das Problem ist gelöst:

```
> Rule[revert](%);
```

```
Rule [revert] does not apply
```

$$\int_{w=a}^{w=b} \frac{u+1}{3(u^2-u+1)} - \frac{1}{3(u+1)} du = \frac{1}{6} \ln(u^2-u+1) \Big|_{w=a..b} + \frac{1}{2} \int_{w=a}^{w=b} \frac{1}{-3ul+ul^2+3} du - \frac{1}{3} \ln(u+1) \Big|_{w=a..b}$$

f) Experimentieren (Undo)

Insbesondere Integrationsprobleme können recht schwer sein, da man mitunter verschiedene Regeln ausprobieren muss, um die gewünschte Vereinfachung zu erzielen. Um das zu ermöglichen, gibt es den Befehl **Undo**, der eine Regelanwendung rückgängig macht.

```
> Int(sqrt(1-x^2), x);
```

$$\int \sqrt{1-x^2} dx$$

```
> Rule[change,x=tan(u)](%) ;
```

```
Creating problem #10
```

```
Applying substitution x = tan(u), u = arctan(x) with dx = (1+tan(u)^2)*du, du = 1/(x^2+1)*dx
```

$$\int \sqrt{1-x^2} dx = \int \sqrt{1-\tan(u)^2} du + \int \sqrt{1-\tan(u)^2} \tan(u)^2 du$$

```
> Undo(%);
```

$$\int \sqrt{1-x^2} dx$$

```
> Rule[change,x=sin(u)](%) ;
```

```
Applying substitution x = sin(u), u = arcsin(x) with dx = cos(u)*du, du = 1/(1-x^2)^(1/2)*dx
```

$$\int \sqrt{1-x^2} dx = u - \int \sin(u)^2 du$$

2. Kurvendiskussion

```
> restart:
```

```
with(Student:-Calculus1):
```

Kurvendiskussion leicht gemacht:

```
> f := x -> (x^2 - 4) / (x + 3);
```

$$f := x \rightarrow \frac{x^2 - 4}{x + 3}$$

```
> Roots( f(x), x );
```

$$[-2, 2]$$

```
> ExtremePoints( f(x), x );
```

$$[-3 - \sqrt{5}, \sqrt{5} - 3]$$

```
> InflectionPoints( f(x), x );
```

$$[-3]$$

```
> Asymptotes( f(x), x );
```

$$[y = x - 3, x = -3]$$

```
> CriticalPoints( f(x), x );
```

$$[-3 - \sqrt{5}, -3, \sqrt{5} - 3]$$

3. Visualisierung mathematischer Verfahren/Sachverhalte

```
> restart;  
with(Student:-Calculus1):
```

a) Integration

Riemann-Summe:

```
> ApproximateInt(sin(x), x=0..20, method=midpoint, output=plot);
```


An Approximation of the Integral of

$$f(x) = \sin(x)$$

on the Interval $[0, 20]$

Using a Midpoint Riemann Sum

Approximate Value: .5919179382



Simpson-Methode: Approximation des Integrals durch quadratische Funktionen:

```
> ApproximateInt(sin(x), x=0..20, method=simpson, output=plot);
```


An Approximation of the Integral of

$$f(x) = \sin(x)$$

on the Interval $[0, 20]$

Using Simpson's Rule

Approximate Value: .5919179382



Animation zur Riemann-Integration:

```
> ApproximateInt(sin(x), x=0..3*Pi/2, output=animation);
```


An Approximation of the Integral of

$$f(x) = \sin(x)$$

on the Interval $[0, 3/2\pi]$

Using a Midpoint Riemann Sum

Approximate Value: 1.000000000



Zufällige Wahl der Verfeinerung:

```
> ApproximateInt(sin(x), x=0..4*Pi, partition = random[2], subpartition=width,  
refinement=random, output=animation, iterations=50);
```


An Approximation of the Integral of

$$f(x) = \sin(x)$$

on the Interval $[0, 4\pi]$

Using a Midpoint Riemann Sum

Approximate Value: $-8.091166047 \times 10^{-15}$



Undefiniertes bestimmtes Integral:

```
> ApproximateInt(1/(x^2 - 2), x=-2..2, partition = random[2], subpartition=width,  
refinement=random, output=animation, iterations=50);
```


An Approximation of the Integral of

$$f(x) = 1/(x^2-2)$$

on the Interval $[-2, 2]$

Using a Midpoint Riemann Sum

(Integral is Undefined)



b) Newton-Verfahren

```
> NewtonsMethod(sin(x) + 1, x = 1);
```

-1.619017361

```
> NewtonsMethod(sin(x) + 1, x = 2, output = sequence);
```


2, 6.588037826, 5.225039135, 4.962948756, 4.837009302, 4.774618375

```
> NewtonsMethod(x^3 - x, x = -0.432, view = [-2..1, DEFAULT], output = plot);
```

5 Iterations of Newton's Method Applied to

$$f(x) = x^3 - x$$

with Initial Point $x = -0.432$




— f(x)
— Tangent lines

```
> NewtonsMethod(x^3 - x, x = 2, view = [0..3, DEFAULT], output = plot);
```

5 Iterations of Newton's Method Applied to

$$f(x) = x^3 - x$$

with Initial Point $x = 2$




```
> NewtonsMethod(x^2 + x + 1, x = 2, output = plot);
```

5 Iterations of Newton's Method Applied to

$$f(x) = x^2 + x + 1$$

with Initial Point $x = 2$



c) Rotationskörper

```
> VolumeOfRevolution(x^2 + 1, x=0..1);
```

$$\frac{28\pi}{15}$$

```
> VolumeOfRevolution(x^10 + 1, x^2 + 1, x=0..1);
```

$$-\frac{736\pi}{1155}$$

```
> VolumeOfRevolution(sin(x) + 1, x=0..3, output=integral);
```

$$\int_0^3 \pi (\sin(x) + 1)^2 dx$$

```
> VolumeOfRevolution(sin(x) + 1, x=2*Pi..3*Pi, output=integral, axis=vertical);
```


$$\int_{2\pi}^{3\pi} 2\pi x (\sin(x) + 1) dx$$

```
> VolumeOfRevolution(cos(x) + 1, x=0..4*Pi, output=plot);
VolumeOfRevolution(cos(x) + 3, sin(x) + 2, x=0..4*Pi, output=plot);
VolumeOfRevolution(sin(x) + 1, x=2*Pi..3*Pi, output=plot, axis=vertical);
```

The Volume of Revolution Around the Horizontal Axis of

$$f(x) = \cos(x)+1$$

on the Interval $[0, 4\pi]$




The Volume of Revolution Around the Horizontal Axis Between

$$f(x) = \cos(x)+3$$

and

$$g(x) = \sin(x)+2$$


on the Interval $[0, 4\pi]$



The Volume of Revolution Around the Vertical Axis of

$$f(x) = \sin(x)+1$$


on the Interval $[2\pi, 3\pi]$



d) Kurvendiskussion noch leichter gemacht


```
> FunctionChart( (x^2 - 4) / (x + 3) );
```

The Chart of
 $f(x) = (x^2-4)/(x+3)$
on the Interval $[-10, 10]$



```
> FunctionChart( sin(x), x=0..2*Pi );
```

The Chart of
 $f(x) = \sin(x)$
on the Interval $[0, 2\pi]$



```
> FunctionChart( (x^2-2) / (x^3-x+1), x=-6..6 );
```

The Chart of
 $f(x) = (x^2-2)/(x^3-x+1)$
on the Interval $[-6, 6]$

